

# A Blockchain Based Decentralised Booking System

Naipeng Dong<sup>1</sup>, Guangdong Bai<sup>2</sup>, Lung-Chen Huang<sup>3</sup>, Edmund Kok Heng Lim<sup>4</sup>  
and Jin Song Dong<sup>5</sup>

<sup>1,3,4,5</sup>*School of Computing, National University of Singapore, COM1, 13 Computing Drive, 117417, Singapore*

<sup>2,5</sup>*School of Information and Communication Technology, Griffith University, N44 2.28, 170 Kessels Road Nathan, QLD, 4111, Australia*

*E-mail:* <sup>1</sup>*dcsdn@nus.edu.sg*, <sup>2</sup>*g.bai@griffith.edu.au*, <sup>3</sup>*lungchenhuang@u.nus.edu*, <sup>4</sup>*e0335737@u.nus.edu*,

<sup>5</sup>*dcsdjs@nus.edu.sg*

## Abstract

Blockchain technology has rapidly emerged as a decentralised trusted network to replace the traditional centralised intermediary. Especially, the smart contracts that are based on blockchain allow users to define the agreed behaviour among them, the execution of which will be enforced by the smart contract. Based on this, we propose a decentralised booking system that uses the blockchain as the intermediary between hoteliers and travellers. The system enjoys the trustworthiness of blockchain, improves efficiency and reduces the cost of the transitional travel agencies. The design of the system has been formally modelled using the CSP# formal language and has been formally verified using the model checker PAT. We have implemented a prototype decentralised booking system based on the Ethereum ecosystem.

## 1 Introduction

The blockchain technology has rapidly emerged in recent years (Nakamoto (2019), Buterin (2019), Swan (2015)), especially when the concept of smart contract was first introduced to and later relied on the technology (Morris (2019), Cachin (2019), Schwartz *et al.* (2014)). Once the smart contract code is deployed to the blockchain, it can be executed by any computer node that keeps the same historical record of transactions as other nodes. This makes it difficult to be compromised with a single node on the network, unlike centralized platforms that could be easily breached and prone to the failure of single point. However, decentralized platforms are still in its infancy. Although some communities and companies have proposed some use cases for the logistics (DHL (2019), Marr (2019)) and insurance industry (CBInsights (2019), Sarasola (2019)), its use cases are lacking in different domains. To give a more concrete example of using the smart contract, we design a fundamental architecture of a decentralized hotel booking system.

Most travelers would have hard times finding a hotel room for their next journey. They would possibly browse through pages of entries on online travel agencies such as *Booking.com* or *Agoda.com*. As we can imagine, it is time-consuming that each time a search result appears on

the screen, they have to delve into the deals one by one, which could be overlapping in the last search results. In order to alleviate this monotonous experience, we shall leverage the blockchain technology and allow our search requests to be deployed as smart contracts such that the process of discovery is left to the decentralized system, where hoteliers can easily match their rooms with the criteria required by a traveler.

Our approach is to provide users with an interface where they can draft their booking request with the requirement of a hotel room in a domain-specific language, which is similar to a real contract in a human-readable form. Thereafter, the interface compiles the request into a machine-readable form and injects some predefined functions. Later, the user can deploy the request onto the blockchain. Once the request is visible to other nodes, hoteliers can propose offers by invoking a function in the request. When the user is notified of new proposals, the interface automatically starts a selection process of the proposals depending on the request criteria from the user, showing the matched results. At the end, the user and one of the hoteliers seal a deal.

To ensure the correctness of the design, we model the system using the CSP# (Communicating Sequential Programs sharp) formal language (Sun *et al.* (1999)), as CSP# integrates the high-level modeling operators of CSP with low-level procedural codes in C# language and supports custom data structures which is extensively used in our model to represent the behaviors of the blocks and the blockchain. We verify the design using the model checker PAT (Process Analysis Toolkit, available at [pat.comp.nus.edu.sg](http://pat.comp.nus.edu.sg)) which supports the CSP# models as input. The formal model is comprehensive, containing not only the behaviour of the smart contract, the hoteliers and the travellers, but also contains the blockchain mechanism.

In addition, we have implemented a prototype booking system based on the Ethereum ecosystem (Ethereum (2019)), and tested the prototype system on a Ethereum Virtual Machine (EVM). The booking system contains three components: the hotelier, the traveler and the smart contract. The EVM mainly contains the EVM nodes, the miner nodes. We use *Truffle* to compiler the smart code and deploy the compiled bytecode to the EVM nodes. In addition, we installed a web server to test the hotelier and traveller websites.

## 2 Background

In this work, we use the CSP# as the language to formally model the design. This language is supported by the automatic model checker PAT. Thus, we use PAT to automatically verify a set of desired properties of our design. The syntax of this language and the model checker PAT are briefly introduced in the first section. After verifying the correctness of the design, we implemented the design based on EVM, which is described in the second part.

### 2.1 CSP# and PAT

The CSP# is a rich modelling language that contains both high level modelling operators in the traditional CSP language and programmer-favored low-level constructs like variables, arrays, if-then-else, while, etc.. It offers great flexibility to model systems with complicated structures, like blockchain. A system is modelled as a process in CSP#. And the operators in CSP# is defined as follows.

h

Process $P, Q ::=$	
$Stop$	– deadlock
$Skip$	– termination
$[b]P$	– state guard
$e \rightarrow P$	– event prefixing
$e\{program\} \rightarrow P$	– data operation prefixing
$c?d \rightarrow P(d)$	– channel input
$c!d \rightarrow P$	– channel output
$P; Q$	– sequence
$P \sqcap Q$	– internal choice
$P \square Q$	– external choice
$if\ b\ then\ P\ else\ Q$	– conditional branch
$P \parallel Q$	– synchronous
$P \parallel\!\!\!\parallel Q$	– asynchronous

**Figure 1** CSP# Syntax

where  $P$  and  $Q$  are *processes*,  $b$  is a condition,  $e$  is a simple event, *program* is a block of code that is atomically executed and  $c$  is a synchronized communication channel.

The deadlock process is *Stop*, meaning the process does absolutely nothing. Process *Skip* means the process terminates immediately. Process  $[b]P$  is a guarded process - the process behaves as  $P$  if  $b$  is satisfied. Process  $e \rightarrow P$  is event prefixing - the process performs event  $e$  (a simple event is a name for representing an observation) and then behaves as  $P$ . Process  $e\{program\} \rightarrow P$  is similar to event prefixing. The difference is that the  $\{program\}$  attached with event  $e$  allows us to write assignments which update global variables. Process  $c?d \rightarrow P(d)$  is channel input - the process reads a value  $d$  from channel  $c$ , thus  $d$  is known in the subsequent process  $P$ . Process  $c!d \rightarrow P$  is channel output - the process outputs value  $d$  in channel  $c$ . Process  $P; Q$  concatenates two processes  $P$  and  $Q$  sequentially. Process  $P \sqcap Q$  is internal choice - the process either chooses  $P$  or  $Q$  to execute. If  $P$  performs an event first, then  $P$  takes control; Otherwise  $Q$  takes control.  $P \square Q$  is external choice. Differing from the internal choice, the external choice is resolved by the environment. Process *if b then P else Q* is straight forward: if  $b$  is true, the process behaves as  $P$ ; Otherwise behaves as  $Q$ . Both  $P \parallel Q$  and  $P \parallel\!\!\!\parallel Q$  model parallel processes  $P$  and  $Q$ . The difference is that the former one requires  $P$  and  $Q$  to synchronise on the shared events whereas in the latter process,  $P$  and  $Q$  interleave.

PAT is a self-contained framework which supports simulating and reasoning and verifying concurrent systems. It has user friendly interfaces, featured model editor and animated simulator. Most importantly, PAT implements various model checking techniques catering for different properties such as deadlock-freeness, reachability and LTL properties that are useful for our system verification. To achieve good performance, advanced optimization techniques are implemented in PAT, e.g. partial order reduction, symmetry reduction, process counter abstraction, parallel model checking. So far, PAT has 4370+ registered users from 1341+ organizations in 150 countries and regions.

## 2.2 Ethereum


The Ethereum ecosystem serves as a runtime environment for smart contracts. It provides a language *Solidity* to write smart contracts. The smart contracts need *Gas* to be executed. And gas can only be bought using the cryptocurrency *Ether* Provided by Ethereum. It mainly involves the following parts (Ethereum (2019), Modi (2019), Buterin (2019)):

- EVM nodes: The EVM runs smart contract bytecode. Each node runs an instance of the EVM. Once written, the contracts have to be deployed on the EVM. Users can interact with the contracts that are on the EVM. The EVM node is a host for the Ethereum network. There can be many EVM nodes serving one network, in that sense the Ethereum network is decentralised.
- Mining nodes: The mining nodes (a.k.a. miners) form the basic blockchain. Each block contains data and smart contract code, and holds a copy of the blockchain. The mining nodes are required to verify and add blocks to the chain and in the process get paid in Ether. Every transaction requires Gas to run. Gas is the measure of the amount of Ether that is required to run.
- Externally owned account: It represents a user. When a new account is created, a private public key pair is created. The Externally owned account is the public key of the account.
- Wallet: It is a software that is associated to user accounts. One user may have more than one account in one wallet. The wallet is an important component as it is the software that tracks the amount of Ether one has. Mining nodes also require a wallet to store the ether earned. Most wallets will show transactions made; these transactions are identified by their hash values.
- Smart contracts: Smart contracts are programs that exist on the EVM. They can accept inputs and based on the inputs, can produce an action whether it is to read a value from the blockchain or to write a value to the blockchain. Smart contracts are the key to interacting with the blockchain programmatically.
- User interface: It refers to any interface that is interacting with a smart contract on the EVM. User interfaces together with smart contracts are known as DApps or decentralised apps.

## 3 System design

Using the Ethereum environment, we will not need to consider the blockchain in the system design, as the EVM provides interfaces to communicate with the underlying blockchain. What we need in the system design is mainly three parts: the smart contract, the traveler behaviour and the hotelier behaviour. To more clearly illustrate the workings of our system, the system sequence diagram in Figure 2 describes how travelers and hoteliers interact on the decentralized booking platform.

Once the smart contract is deployed on the EVM, a traveler can put a request to the smart contract and then listens to the contract for any response. A Hotelier listens to the contract and once discovers a request he proposes his offer. After that the hotelier listens to the contract for response. When the traveler accepts the offer, he settles the request and then pay the deposit.



**Figure 2** Overview of the system design

On receiving the deposit payment, the hotelier change the room into unavailable and informs the smart contract. Finally, the traveler disable the request.

However, the entire system, including the blockchain, contains three roles *user*, *miner*, and the smart contract. And we use model the entire system in the formal verification as it is more comprehensive.

- A user could be a traveler or hotelier who sends a transaction by invoking a function that includes the address of the user, function name, gas, and gas price. A traveler can invoke Fetch to acquire the latest set of proposals submitted by hoteliers, or to Settle for a specific proposal. A hotelier can invoke Propose to send their proposals to the request.
- Miners form the basic blockchain network. Each miner has its own transaction pool *TxPool* that continuously receives a new transaction with gas greater than zero. Moreover, *TxBlock* to be executed by a miner also finds the new transactions with more than or equal to a specified transaction *gasPrice* from the pool.
- The smart contract is named as the Request. The request deployed on the blockchain is a smart contract that accepts proposals and can be invoked by the users.

## 4 System modelling

First of all, we define a list of constants that are used in the model. Then we divid the model into two parts: the *user* behaviour in Section 4.1 and the *Miner* behaviour in Section 4.2. The miner part includes the behaviour of the EVM nodes, the mining nodes, and the smart contract, since these behaviours are closely connected.

```
enum {off, on, switchOp, proposeOp, fetchOp, settleOp, proposedFrom,
      Function, ProposeFunction, settled, na, proposed, SwitchFunction,
      FetchFunction, SettleFunction, newcomer};
```

**Figure 3** Constants declaration

#### 4.1 user

A user has an address *addr*. The user would like to execute *function* and would like to pay a fund *val* plus some gas amount *gas* with price *gasPrice*. To do so, the user broadcasts the transaction to all miners. The reason that the user needs to specify the gas amount and gas price is that the gas price is changing according to the demand and supply; to be clear on how much the user pays the miner for executing the function, the user specifies the current gas price and the amount of gas the user would like to pay. For simplicity, we assume the user sends the transactions to miners in order, as this will not affect the results we would like to check. The process of sending a transaction to miner *i* is modelled as *TxBrcdst*(*i*, *addr*, *val*, *gas*, *gasPrice*, *function*). When the miner index *i* is correct (i.e., *i* is smaller than the maximum number of miners), this process sends a transaction to the *i*-th miner with the parameters *addr*, *val*, *gas*, *gasPrice*, *function*, and then continues to send the transaction to the next miner. If the miner index reaches the top, meaning that the user has sent to every miner, then the process terminates with *Skip*.

```
User(addr, val, gas, gasPrice, function) =
    TxBrcdst(0, addr, val, gas, gasPrice, function);

TxBrcdst(iter, addr, val, gas, gasPrice, function) =
    atomic{
        if (iter < M) {
            mnet[iter]!addr.val.gas.gasPrice.function ->
                TxBrcdst(iter+1, addr, val, gas, gasPrice, function) }
        else {
            Skip } };
```

**Figure 4** The user process

#### 4.2 Miner

The miner has two processes running in parallel: process *TxPool*(*i*) which receives transactions and puts them into a pool, and process *TxInsert* which includes the transactions in the pool into a block and initiates the mining.

The miner *i* receives the transactions in the process *TxPool*(*i*). On receiving a transaction with parameters, the miner first checks whether the sender i.e., the user with address *addr* has enough funds and gas, if so, the miner inserts the transaction into his transaction pool *TxInsert*(*i*, *addr*, *val*, *gas*, *gasPrice*, *function*), otherwise, the miner keeps waiting for another transaction, i.e., returning back to process *TxPool*(*i*).

```

TxPool(i) =
  mnet[i]?addr.val.gas.gasPrice.function ->
  if (userWallet[i][addr] >= (val + (gas * gasPrice)) && gas > 0)
    {TxInsert(i, addr, val, gas, gasPrice, function) }
  else {TxPool(i) };

```

**Figure 5** The transaction pool process

When inserting a transaction into a pool of miner  $i$ , the miner first check whether the pool is full, i.e., the current transactions in  $i$  ( $PoolPtr[i]$  is smaller than the max  $poolSize$ ). If not, the miner receives the transaction in the pool by updating the address, the value, the gas, the gasPrice and the function to the pool of miner  $i$ . At the end, we update the pointer of the transaction pool to the next empty position. After updating the transaction pool, the miner sends a single of *newcommer* to the channel *datach* for miner  $i$  to initiate the block forming. Finally the process goes back to  $TxPool(i)$  to listen to a new transaction. Note that before updating the transaction pool, we double check whether the current empty pool position is really empty by checking whether the gas for the current position is 0. If the current position is not empty, the process move to the next position and tries to insert the transaction to the pool. If the transaction pool is full, the transaction pointer is reset to 0 and then the transaction is inserted into the transaction pool, meaning to form transactions for the next block. We notice that if there are more than the maximum amount of transactions received in the pool before a block is formed, the pool will be reset to new ones. In our model, the total transactions are less than the maximum transitions in a pool, so this will not happen.

```

TxInsert(i, addr, val, gas, gasPrice, function) =
  if (poolPtr[i] < poolSize) {
    if (poolTxGas[i][poolPtr[i]] == 0) {
      txReceive{
        var ptr = poolPtr[i];
        poolTxAddr[i][ptr] = addr;
        poolTxVal[i][ptr] = val;
        poolTxGas[i][ptr] = gas;
        poolTxGasPrice[i][ptr] = gasPrice;
        poolTxFunction[i][ptr] = function;
        poolPtr[i]++;} ->
      datach[i]!newcomer ->
      TxPool(i) }
    else {
      next{poolPtr[i]++;} ->
      TxInsert(i, addr, val, gas, gasPrice, function) } }
  else {
    next{poolPtr[i] = 0;} ->
    TxInsert(i, addr, val, gas, gasPrice, function)};

```

**Figure 6** The transation insertion process

The process  $TxBlock(i)$  include available transitions into a block. When the number of transactions in the pool is smaller than the pool size and the channel for receiving transactions is not empty, the process receives a signal that a new transaction has been inserted into the transaction pool. Otherwise the process sets the pointer  $inPoolPtr(i)$  to 0 and goes back to the process  $TxBlock(i)$ .

```

TxBlock(i) =
  if (inPoolPtr[i] < poolSize) {
    if (!call(cempty, datach[i])) {
      datach[i]?newcomer ->
      if (poolTxGas[i][inPoolPtr[i]] == 0) {
        if (blkSize[i] >= 1) {
          Miner(i, 0)}
        else { TxBlock(i) } }
      else {
        if (blkTotalGas[i] + poolTxGas[i][inPoolPtr[i]] <= blkGasLimit) {
          if (poolTxGasPrice[i][inPoolPtr[i]] >= gasPriceCondition) {
            txInclude{
              var pIter = inPoolPtr[i];
              var bIter = blkSize[i];
              blkTxAddr[i][bIter] = poolTxAddr[i][pIter];
              blkTxVal[i][bIter] = poolTxVal[i][pIter];
              blkTxGas[i][bIter] = poolTxGas[i][pIter];
              blkTxGasPrice[i][bIter] = poolTxGasPrice[i][pIter];
              blkTxFunction[i][bIter] = poolTxFunction[i][pIter];
              blkTotalGas[i] = blkTotalGas[i] + poolTxGas[i][pIter];
              poolTxAddr[i][pIter] = 0;
              poolTxVal[i][pIter] = 0;
              poolTxGas[i][pIter] = 0;
              poolTxGasPrice[i][pIter] = 0;
              poolTxFunction[i][pIter] = 0;
              inPoolPtr[i]++;
              blkSize[i]++;} ->
            TxBlock(i)}
          else {nextTx{inPoolPtr[i]++;} ->
            TxBlock(i)}}
        else if (blkSize[i] >= 1) {Miner(i, 0)}
        else {nextTx{inPoolPtr[i]++;}->
          TxBlock(i)}}}
      else if (blkSize[i] >= 1) {Miner(i, 0)}
      else { TxBlock(i) } }
    else {nextTx{inPoolPtr[i] = 0;} -> TxBlock(i)};
  }

```

**Figure 7** The process of including transactions into a block

On receiving the signal, the miner checks whether the gas for this transaction is 0 indicating there is no transaction in the pool.

- If there is no transaction in the pool and the block size is bigger than 1, meaning all the transactions have been included the block, the miner starts mining block by in the calling



$Miner(i, 0)$ . If there is no transaction in the pool and the block size is smaller than 1, meaning there is no transaction in the block, then the process goes back to  $TxBlock(i)$ .

- If there are transactions in the pool, the miner checks whether the total gas for the current block reaches the top limit and checks whether the current transaction in the pool provides enough gas to execute the function. If so, the miner include the transaction in the block, by copying the address, value, gas, gas price and function into the block. In addition, the miner update the total gas of the block and set the position of the transaction in the pool to be 0 indicating that the transaction in the pool has been included into the block. Furthermore, the pointer in the pool will move to the next and the size of the block increase by 1. After updating the current transaction in to pool, the process goes back to  $TxBlock(i)$  to include the next transaction into the pool, until all transactions have been included, i.e., the pointer  $inPoolPtr[i]$  is bigger than the  $poolSize$ . If the current transaction in the pool fails to provide enough gas, the miner simply ignores the transaction and moves to the next one. If the total gas has reached the top limit and there is transaction in the block, the miner starts mining. If the total gas reaches the top, but somehow there is no transaction in the block, the process goes back.

The mining process starts with a checking on whether the block index is smaller than the block size of the miner  $i$ . If so, the miner locks up the total value and executes the functions in the block.

```
Miner(i, iter) = if (iter < blkSize[i]) {LockUp(i, iter)}
               else { BlkUpdate(i, 0)};
```

**Figure 8** The miner process

To lock up the total value, the miner  $i$  calculates the amount to lock by adding the fund value and the gas value as the locked total amount. The remaining amount for the user with address  $addr$  is also calculated by deducting the locked amount. Once the amount for a transaction  $iter$  is locked, the miner executes the transaction by calling  $TxExec(i, iter)$ .

```
LockUp(i, iter) =
  weiLockUp{
    var addr = blkTxAddr[i][iter];
    var val = blkTxVal[i][iter];
    var gas = blkTxGas[i][iter];
    var gasPrice = blkTxGasPrice[i][iter];
    var lockedTotal = val + (gas * gasPrice);
    userWallet[i][addr] = userWallet[i][addr] - lockedTotal;
    lockedWallet[i] = lockedTotal;} ->
  TxExec(i, iter);
```

**Figure 9** The lock up process

To execute a transaction *iter*, the miner reads in the function in the transaction. There are four functions provided in the smart contract: *SwitchFunction*, *ProposeFunction*, *SettleFunction* and *FetchFunction*, which are defined as constants.

- When *SwitchFunction* is called, the miner checks whether the user who calls the function is the contract owner. If so, the miner executes the transaction and updates the gas with price of *UPDATE*; Otherwise returns false to indicate that the function is not successfully executed.
- When *ProposeFunction* is called, the miner checks whether the contract is switched on. If so, the miner updates the gas with price of *UPDATE* and executes the transaction; Otherwise returns false.
- When *SettleFunction* is called, the miner checks whether the user who calls the function is the contract owner. If so, the miner executes the transaction and update the gas with price of *UPDATE + TRANSFER*; Otherwise returns false. Differing from the previous two functions, this function costs more, including the updating fee and the transferring fee.
- When *FetchFunction* is called, the miner checks whether the user who calls the function is the contract owner. If so, the miner executes the transaction and update the gas with the price of *FETCH*.

```

TxExec(i, iter) =
case {
  blkTxFunction[i][iter] == SwitchFunction:
    if (contractOwner == blkTxAddr[i][iter]) {
      GasConsume(i, iter, UPDATE) || Execution(i, iter)}
    else { LockedReturn(i, iter, false) }
  blkTxFunction[i][iter] == ProposeFunction:
    if (contractSwitch[i] == on) {
      GasConsume(i, iter, UPDATE) || Execution(i, iter)}
    else {LockedReturn(i, iter, false)}
  blkTxFunction[i][iter] == SettleFunction:
    if (contractOwner == blkTxAddr[i][iter]) {
      GasConsume(i, iter, UPDATE + TRANSFER) || Execution(i, iter)}
    else {LockedReturn(i, iter, false)}
  blkTxFunction[i][iter] == FetchFunction:
    if (contractOwner == blkTxAddr[i][iter]) {
      GasConsume(i, iter, FETCH) || Execution(i, iter)}
    else {LockedReturn(i, iter, false)}};

```

**Figure 10** The overall process of transaction execution

Before executing the functions in the transaction, the miner first checks whether the transaction has enough gas. If not, the execution fails, modelled in the process *ExecFail(i, iter)*. If there is enough gas, the miner updates the gas by deducting the amount (specified in the parameter according to different functions) from the locked wallet and added to the miner's wallet. Then the miner imitates the execution, modelled by the synchronised event *consumed*.

```

GasConsume(i, iter, opcode) =
  if (estimatedGas[i][iter] + opcode > blkTxGas[i][iter]) {
    ExecFail(i, iter)}
  else {
    consuming{
      var gasPrice = blkTxGasPrice[i][iter];
      var price = opcode * gasPrice;
      minerCoinbase[i] = minerCoinbase[i] + price;
      lockedWallet[i] = lockedWallet[i] - price;
      estimatedGas[i][iter] = estimatedGas[i][iter] + opcode;} ->
      consumed ->
      Skip};

ExecFail(i, iter) =
  lockedReset{ lockedWallet[i] = 0;} ->
  BlkDetect(i, iter, false);

```

**Figure 11** The gas update process

```

Execution(i, iter) = consumed ->
  execution{
    pendUId[i][iter] = txUId;
    pendFromAddr[i][iter] = blkTxAddr[i][iter];
    pendToAddr[i][iter] = contractAddr;
    pendOp[i][iter] = blkTxFunction[i][iter];
    pendField[i][iter] = na;
    pendState[i][iter] = 1;
    pendValue[i][iter] = blkTxVal[i][iter];
    txUId++;} ->
    LockedReturn(i, iter, true);

LockedReturn(i, iter, success) =
  returnGas{
    var addr = blkTxAddr[i][iter];
    var return = (blkTxGas[i][iter] - estimatedGas[i][iter]) * blkTxGasPrice[i][iter];
    userWallet[i][addr] = userWallet[i][addr] + return;
    lockedWallet[i] = 0; } ->
    BlkDetect(i, iter, success);

BlkDetect(i, iter, success) =
  if (call(cempty, bnet[i])) {Miner(i, iter+1)}
  else {bnet[i]?j.newBlockNum.blockid -> BlkAppend(i, j, newBlockNum, blockid)};

```

**Figure 12** The actual execution process

Once the event *consumed* is synced, the miner executes the function. We model the execution of a function *Execution(i, iter)* by giving the transaction a unique ID and storing the transaction owner, contract address, transaction function and transaction fee in a set of arrays to indicate that the function of the smart contract in that transaction has been executed. Thus, the process

returns *true* by calling process *LockedReturn(i, iter, true)*. Then the miner returns the remaining locked gas to the user. Before the miner executes the next transaction, she checks whether there is any block from another miner. If so, the miner chooses to execute the first block she receives, modelled in process *BlkDetect(i, iter, success)*.

Recall that the miner process starts with a check on whether the block index is smaller than the block size of the miner *i*. If not, i.e., the transaction pool for the miner is full, the miner updates the block by rewarding the miner and broadcasting the block as in process *BlkUpdate(i, iter)*.

```

BlkUpdate(i, iter) =
  if (iter < blkSize[i]) {
    update{
      block[blockUid][iter] = pendUid[i][iter];
      txFromAddr[pendUid[i][iter]] = pendFromAddr[i][iter];
      txToAddr[pendUid[i][iter]] = pendToAddr[i][iter];
      txOp[pendUid[i][iter]] = pendOp[i][iter];
      txField[pendUid[i][iter]] = pendField[i][iter];
      txState[pendUid[i][iter]] = pendState[i][iter];} ->
      BlkUpdate(i, iter+1)}
  else {reward{minerCoinbase[i] = minerCoinbase[i] +
    succAppendPrice; rewardCount++;} ->
    BlkBrdcst(i, 0, blockNum[i], blockUid)};

```

**Figure 13** The block update process

To broadcast a block, the miner sends the block number and block ID to other miners. Once this has been done, the miner increase the block ID by 1 and append the block to her chain (process *BlkAppend(i, i, newBlockNum, blockid)*).

```

BlkBrdcst(i, iter, newBlockNum, blockid) =
  if (i == iter && iter < M) {BlkBrdcst(i, iter+1, newBlockNum, blockid)}
  else if (iter < M) {
    bnet[iter]!i.newBlockNum.blockid ->
    BlkBrdcst(i, iter+1, newBlockNum, blockid)}
  else {
    updateBlockUid{blockUid++;} ->
    BlkAppend(i, i, newBlockNum, blockid)};

```

**Figure 14** The block broadcasting process

```

BlkAppend(i, j, newBlockNum, blockid) =
  append{chain[i][newBlockNum] = blockid;
    blockNum[i]++;
    blkSize[i] = blkSize[j];} ->
  ChainUpdate(i, 0, blockid);

```

**Figure 15** The block appending process

In process *BlkAppend*(*i, i, newBlockNum, blockid*), the minder appends the latest block and updates the state in the blockchain.

The process of updating the blockchain actually updates the results of the functions. When the operation is *FetchFuction*, we add a *fetch* event to denote it; When the operation is *SiwtchFunction*, the process switch the smart contract either from the current state *on/off* to the alternative *off/on*; When the operation is *ProposeFunction*, the process executes a *propose* event, where the proposal initiator and the proposal state are recorded; When the operation is *SettleFunction*, the process executes a *settle* event, which records the one who settle the deal and the corresponding transaction state. Once an operation is finished, the process goes back to itself for the operation in the next transaction. Note that the process first check whether all the transaction has been updated. If so, the process reset the transaction container in process *Reset*(*i, 0*).

```
ChainUpdate(i, iter, blockid) =
  if (iter < blkSize[i]) {
    case {
      txOp[block[blockid][iter]] == FetchFunction:
        fetch ->
        ChainUpdate(i, iter+1, blockid)
      txOp[block[blockid][iter]] == SwitchFunction:
        switch{var id = block[blockid][iter];
          if (contractSwitch[i] == off) {contractSwitch[i] = on;}
          else {contractSwitch[i] = off;}} ->
        ChainUpdate(i, iter+1, blockid)
      txOp[block[blockid][iter]] == ProposeFunction:
        propose{var id = block[blockid][iter];
          var addr = txFromAddr[id];
          var data = txState[id];
          var ptr = proposalPtr[i];
          proposalFrom[i][ptr] = addr;
          proposal[i][ptr] = data;
          proposalPtr[i]++; } ->
        ChainUpdate(i, iter+1, blockid)
      txOp[block[blockid][iter]] == SettleFunction:
        settle{var id = block[blockid][iter];
          var addr = txFromAddr[id];
          var data = txState[id];
          settledWith[i] = data;
          userWallet[i][data] = userWallet[i][data] + txValue[id]; } ->
        ChainUpdate(i, iter+1, blockid)}}
    else {Reset(i, 0) };
```

**Figure 16** The chain updating process

In the reset process, the miner clear each transaction container by setting the values into 0 shown as follows. Once all the transactions in a block is cleared, the miner clear the block by setting the gas and size into 0.

```

Reset(i, iter) =
  if (iter < blkSize[i]) {clearTx{
                                pendUid[i][iter] = 0;
                                pendFromAddr[i][iter] = 0;
                                pendToAddr[i][iter] = 0;
                                pendOp[i][iter] = 0;
                                pendField[i][iter] = 0;
                                pendState[i][iter] = 0;
                                pendValue[i][iter] = 0;
                                blkTxAddr[i][iter] = 0;
                                blkTxVal[i][iter] = 0;
                                blkTxGas[i][iter] = 0;
                                blkTxGasPrice[i][iter] = 0;
                                blkTxFunction[i][iter] = 0;
                                estimatedGas[i][iter] = 0;} ->
                                Reset(i, iter+1)}
  else {clearBlk{
            blkTotalGas[i] = 0;
            blkSize[i] = 0;} ->
        TxBlock(i)};

```

**Figure 17** The reset process

### 4.3 The overall process

In summary, the proposed blockchain based booking system can be modelled as the *user* process and the miner process running in parallel. And the miner process is the transaction pool process *TxPool* and the blockchain process *TxBlock* running in parallel.

```

ProposerExecution = User(1, 0, gasLimit, 5, ProposeFunction) |||
                    (|||i:{0..M-1} @ (TxPool(i) ||| TxBlock(i)));

```

**Figure 18** An overall system model with only one user

## 5 System verification

### 5.1 Properties

We verified the following 5 properties as follows.

- **deadlockfree**: No deadlock situation occurs in the system.
- **GasRunOut**: Each transaction in a block has enough gas to be executed to completion by miners.
- **sameBlockNumEventually**: Each miner reaches the same block number eventually.
- **receiveSettlement**: The request owner has settled with some proposer, and thus each miner receives the same transaction.
- **proposalReceived**: Proposals have been accepted by the request.

Each property is formalised as assertions in PAT for automatic formal verification as shown in Table 1.

Property	Formalisation in PAT
deadlockfree	<i>deadlockfree</i>
GasRunOut	<i>estimatedGas</i> [0][0] > <i>gasLimit</i>
sameBlockNumEventually	$\&\&i : \{0..M\} @ \text{blockNum}[i] == 1$
receiveSettlement	$\&\&i : \{0..M\} @ \text{settledWith}[i] == 1$
proposalReceived	<i>proposalPtr</i> [0] >= 1

**Table 1** Formalisation of the property

For each property, we match one execution. The 5 execution are listed as follows:

- ProposerExecution: A user submits proposals to the request.
- ListnerExecution: A user submits proposals to the request, and the request owner listens to the “proposal” event from the request.
- UnavailabeExecution: The request owner turns off the request, and no more proposals are allowed to be accepted by the request.
- NotOwnerExecution: A user who is not the request owner fetches the proposals from the request.
- MultipleUsersExecution: Many users submit their proposals at the same time.
- SettlementExecution: The request owner seals a deal.

The formalisation of the executions is listed in Table 2.

ProposeExecution	<i>User</i> (1, 0, <i>gasLimit</i> , 5, <i>ProposeFunction</i> )     (    $i : \{0..M - 1\} @ (\text{TxPool}(i)     \text{TxBlock}(i))$ );
ListenerExecution	<i>User</i> (1, 0, <i>gasLimit</i> , 5, <i>ProposeFunction</i> )     <i>Listener</i> ( <i>contractOwner</i> )     (    $i : \{0..M - 1\} @ (\text{TxPool}(i)     \text{TxBlock}(i))$ );
UnavailabeExecution	<i>User</i> ( <i>contractOwner</i> , 0, <i>gasLimit</i> , 5, <i>SwitchFunction</i> )     (    $i : \{0..M - 1\} @ (\text{TxPool}(i)     \text{TxBlock}(i))$ );
NotOwnerExecution	<i>User</i> (0, 0, <i>gasLimit</i> , 5, <i>FetchFunction</i> )     (    $i : \{0..M - 1\} @ (\text{TxPool}(i)     \text{TxBlock}(i))$ );
MultipleUsersExecution	(    $i : \{0..C - 1\} @ \text{User}(i, 0, \text{gasLimit}, 5, \text{ProposeFunction})$ )     (    $i : \{0..M - 1\} @ (\text{TxPool}(i)     \text{TxBlock}(i))$ );
SettlementExecution	<i>User</i> ( <i>contractOwner</i> , 200, <i>gasLimit</i> , 5, <i>SettleFunction</i> )     (    $i : \{0..M - 1\} @ (\text{TxPool}(i)     \text{TxBlock}(i))$ );

**Table 2** Caption

Note that the ‘ProposeExecution’ differs from the ‘MultipleUsersExecution’ in the number of users; the executions ‘ProposeExecution’, ‘UnavailabeExecution’, ‘NotOwnerExecution’ and ‘SettlementExecution’ differ in the function parameters (*ProposeFunction*, *SwitichFunction*, *FetchFunction* and *SettelFunction* respectively). The ‘ListenerExecution’ adds one more process to the ‘ProposeExeution’. The additional process *Listener*(*contractOwner*) is defined as follows, which defines an user interface that listens to the state change.

```
Listener(addr) = listen[addr]?function -> listenerReceiving ->
                User(addr, 0, gasLimit, 5, function);
```

The assertions that match the property with the executions are defined as follows:

- Assertion 1: *ProposerExecution* *deadlockfree*
- Assertion 2: *ProposerExecution*  $\models \Box!GasRunOut$
- Assertion 3: *ProposerExecution* *reaches sameBlockNumEventually*
- Assertion 4: *ListenerExecution*  $\models proposalReceived \rightarrow listenerReceiving$
- Assertion 5: *UnavailableExecution*  $\models \Box!proposalReceived$
- Assertion 6: *NotOwnerExecution*  $\models \Box!fetch$
- Assertion 7: *MultipleUsersExecution* *reaches sameBlockNumEventually*
- Assertion 8: *SettlementExecution* *reaches receiveSettlement*

The Assertion 1 state that the execution *ProposerExecution* is deadlockfree. The Assertion 2 ensures that each transaction has enough gas to be executed. The symbol  $\models$  reads as “satisfies”, which states the execution satisfies the claimed LTL property on the right. The symbol “ $\Box$ ” reads as “globally” meaning every state in the system execution shall satisfy the subsequent LTL formula. The Assertion 3 states that in the execution *ProposerExecution*, each miner reaches the same block. The Assertion 4 ensures that if the proposal is received by the smart contract, then the listener must have received the proposal. The Assertion 5 says that if the smart contract is turned off, then the proposal will not be received. The Assertion 6 states that when a user who is not the smart contract owner fetches the proposal, he will never succeed. The Assertion 7 when there are multiple users, the execution can still reach the same block eventually. The Assertion 8 means when the smart contract owner seals a deal, the deal must have been settled.

```
#define C 5; // # of hoteliers or travelers
#define M 2; // # of miners
#define chainSize 5; // The length of each blockchain
#define maxTx 20; // The maximum number of transaction for each block
#define blkGasLimit 20000; // The maximum gas limit for each block, which is
calculated from a bunch of transactions with gas limit
#define channelBufferSize 5;
#define succAppendPrice 1000000; // reward for successfully appending the
valid block
#define contractAddr 9494; // smart contract address
#define INIT 10; // gas consumption in terms of opcode
#define FETCH 10; // gas consumption in terms of opcode
#define TRANSFER 500; // gas consumption in terms of opcode
#define UPDATE 100; // gas consumption in terms of opcode
#define gasLimit 1000; // gas consumption limit per transaction
#define gasPriceCondition 3; // miner's selection of gas price per transaction
#define contractOwner 2; // the smart contract's owner address
#define poolSize 100; // pool size to receive transactions for each miner
#define proposalPoolSize 10; // pool size for received proposals
#define dataSize 100;
```

**Figure 19** The experiment settings



### 5.2 Experiment settings

For the simplicity of verification, we assume the following settings: there are 5 hoteliers or travelers and 2 miners; The length of blockchain is 5 with the maximally 20 transactions in a block. The maximum gas limit for a block is set to be a large number 20000, which is the gas limit for a transaction times the maximum number of transaction in a block. Other settings can be seen as in Figure 19.

### 5.3 Verification results

Our experiment shows the outcome of the assertions from PAT with respect to different configurations of  $C$  users and  $M$  miners.

Assertion	States	Transitions	Time(s)	Result
Assertion 1	8328	18823	1.76	Valid
Assertion 2	13413	30563	2.68	Valid
Assertion 3	285	336	0.04	Valid
Assertion 4	13413	30563	2.76	Valid
Assertion 5	13413	30563	2.73	Valid
Assertion 6	4537	11940	0.78	Valid
Assertion 7	504	521	0.05	Valid
Assertion 8	362	605	0.06	Valid

**Table 3** The assertion results with  $C = 5$  and  $M = 2$

However, when  $C$  and  $M$  is set to be larger, the state space is too large to verify, as the states and transactions increase exponentially, shown in Table 4 and Table 5.

Assertion	States	Transitions	Time(s)	Result
Assertion 1	6774622	68280939	3647.43	Incomplete
Assertion 2	169626	2165266	112.52	Incomplete
Assertion 3	297717	415836	21.45	Valid
Assertion 4	230685	2897974	415.428	Incomplete
Assertion 5	169929	2168902	111.65	Incomplete
Assertion 6	317081	1511802	189.67	Incomplete
Assertion 7	10312	10425	0.98	Valid
Assertion 8	3313604	32332261	1607.98	Incomplete

**Table 4** The assertion results with  $C = 5$  and  $M = 10$

Assertion	States	Transitions	Time(s)	Result
Assertion 1	512033	523985	201.99	Incomplete
Assertion 2	12140	1240978	194.12	Incomplete
Assertion 3	1005567	1034176	659.31	Incomplete
Assertion 4	15906	1500982	124.36	Incomplete
Assertion 5	15972	1507186	130.03	Incomplete
Assertion 6	181892	907271	510.04	Incomplete
Assertion 7	239832	240425	102.25	Valid
Assertion 8	650388	667096	224.03	Incomplete

**Table 5** The assertion results with  $C = 5$  and  $M = 50$

## 6 System implementation

We have implemented a prototype system (available at [https://github.com/naipengdong/Knowledge\\_engineering\\_Review](https://github.com/naipengdong/Knowledge_engineering_Review)). It contains 3 main components:

- Hotel HTML Page (HotelProposals.html)
- Customer HTML Page (requester.html)
- Hotel Smart Contract (Hotel.sol)

Hotel and Customer HTML Page contains web3.js javascript that interacts with the Hotel Smart Contract. Hotel and Customer HTML Page will each have their own account. The Hotel has an account: `0xfdd7aa0b1bcbb77df50b9ef2a5559808d07396d7` and the customer has an account the account: `0x1c5f708395f2c13cb3471ea3e4332b1b3a6408ca`.

### This is the Hotel View!!

JavaScript can change the content of an HTML element:

Manager Set Contract Panel	View Transactions by Account
This is a demonstration.	
Balance	
Block Number	
123	
Package 1 ▼ Set Contract	
---	
Manager Name: Ed	StartBlock
Required Deposit (for damages): 500	EndBlock
Denomination of currency: SGD	Account
Package Price: 2000	Go!
Number of Days: 4	Status:
Number and type of Rooms: 4D2D	---
Early Bird Discount (Quantum): 40	TransactionLog (Input):
Early Bird End Date (DD/MM/YYYY): 12/12/1980	---
Set Value	TransactionLog (RAW):
---	---
Check Contract Balance	
---	
---	
---	

### This is Customer Page

JavaScript can change the content of an HTML element:

Select - View - Purchase - Checkout Packages	View Transactions
This is a demonstration.	
Balance	
Block Number	
123	
Package 1 ▼ Set Contract	
---	
View Package Details	
---	
	StartBlock
	EndBlock
	Account:
	Go!
	Status:
---	
Pay in SGD ▼ Buy Package?	
Check out?	
---	
---	
---	

**Figure 20** The prototype hotel view and customer page

The implementation of the Hotel Contract DApp demonstrates the use case where

1. Hotel Manager is able to set up a contract, offering 2 packages.
2. A customer is able to view said packages.
3. Customer is able to purchase packages.
4. Customer is able to pay with SGD or ETH.
5. Payment will be withheld in Contract.
6. Contract will release payment to Hotel once Customer checks out.
7. Manager and Customer are both able to view past transactions.

## 7 Conclusions

In this work, we designed a decentralised booking system that links the hoteliers and the travellers, based on the blockchain techniques. The blockchain network replaces the traditional centralised travel agencies that are often inefficient and costly. In addition, it makes the hotel booking business more transparent and avoids the price tricks that can be played by the travel agencies. We have formally modelled the system design including the blockchain using a formal language and verifies a set of properties of the system using a model checker. We implemented a prototype system based on Ethereum and tested its feasibility.

## References

- Buterin, V.: Ethereum white paper. Available at <https://github.com/ethereum/wiki/wiki/White-Paper/f18902f4e7fb21dc92b37e8a0963eec4b3f4793a>, visited at 15 June 2019.
- CBIInsights: How Blockchain Could Disrupt Insurance. Available at <https://www.cbinsights.com/research/blockchain-insurance-disruption/>, visited at 15 June 2019.
- Cachin, C: Architecture of the Hyperledger Blockchain Fabric. Available at [https://www.zurich.ibm.com/dccl/papers/cachin\\_dccl.pdf](https://www.zurich.ibm.com/dccl/papers/cachin_dccl.pdf), visited at 15 June 2019.
- DHL Trend Research: Blockchain in Logistics. Available at <https://www.logistics.dhl/content/dam/dhl/global/core/documents/pdf/glo-core-blockchain-trend-report.pdf>, visited at 15 June 2019.
- Ethereum. Available at <https://www.ethereum.org/>, visited at 14 June 2019.
- Marr, B: How Blockchain Will Transform The Supply Chain And Logistics Industry. Available at <https://www.forbes.com/sites/bernardmarr/2018/03/23/how-blockchain-will-transform-the-supply-chain-and-logistics-industry/#50e51e6d5fec>, visited at 15 June 2019.
- Morris, D. Z: Bitcoin is not just digital currency. It's Napster for finance. Available at <http://fortune.com/2014/01/21/bitcoin-is-not-just-digital-currency-its-napster-for-finance/>, visited at 15 June 2019.
- Modi, R: Introduction to Blockchain, Ethererum and Samrt Contracts. Available at <https://medium.com/coinmonks/https-medium-com-ritesh-modi-solidity-chapter1-63dfaff08a11>, visited at 14 June 2019.
- Nakamoto. S.: Bitcoin: A Peer-to-Peer Electronic Cash System. Available at <https://bitcoin.org/bitcoin.pdf>, visited at 15 June 2019.
- Sun, J., Liu, Y., Dong, J.S., Pang, J.: Pat: Towards flexible verification under fairness. In: International Conference on Computer Aided Verification. pp. 709 714. Springer (2009).
- Sarasola. M.R.: So maybe you figured out what blockchain is but what can you do with it? Available at <https://www.willistowerswatson.com/en-SG/insights/2018/06/emphasis-blockchain-use-in-insurance-from-theory-to-reality>, visited at 15 June 2019.
- Schwartz, D., Youngs, N., Britto, A.: The ripple protocol consensus algorithm. Ripple Labs Inc White Paper 5 (2014).
- Swan, M.: Blockchain: Blueprint for a new economy. O'Reilly Media, Inc. (2015)